Pensieve header: One-Co computations in the abc presentation; continues pensieve://2015-08/.

The bracket

On the elements β, a, c, δa, ca, δaa.

Generalities

```
DQ[is___] := (Sort[{is}] === Union[{is}]);
OQ[is___] := OrderedQ[{is}]; (* tests for non-strict ordering;
also true when {is} is {i,i}. *)
```

```
Simp[expr_] := Simplify[expr];
S[β[f_]] := β[Simp[f]];
S[a[i_, j_]] := a[i, j];
S[a[f_, i_, j_]] := a[Simp[f], i, j];
S[c[f_, k_]] := c[Simp[f], k];
S[δa[f_, i_, j_]] := δa[Simp[f], i, j];
S[ca[f_, j_, k_, l_]] := ca[Simp[f], j, k, l];
S[δaa[f_, i_, j_, k_, l_]] := δaa[Simp[f], i, j, k, l];
S[expr_] := expr /.
   (λ_β | λ_a | λ_δa | λ_c | λ_ca | λ_δaa | λ_β1 | λ_a1 | λ_δa1 | λ_c1 | λ_ca1 | λ_δaa1) ⧴ S[λ];
```

```
β[0] := 0;
β /: β[f_] + β[g_] := β[f + g] // S;
β /: g_ * β[f_] := β[g f] // S;
a[0, _, _] := 0;
a /: a[f_, j_, k_] + a[g_, j_, k_] := a[f + g, j, k] // S;
a /: g_ * a[f_, j_, k_] := a[g f, j, k] // S;
c[0, _] := 0;
c /: c[f_, j_] + c[g_, j_] := c[f + g, j] // S;
c /: g_ * c[f_, j_] := c[g f, j] // S;
δa[0, _, _] := 0;
δa /: δa[f_, j_, k_] + δa[g_, j_, k_] := δa[f + g, j, k] // S;
δa /: g_ * δa[f_, j_, k_] := δa[g f, j, k] // S;
ca[0, _, _, _] := 0;
ca /: ca[f_, j_, k_, l_] + ca[g_, j_, k_, l_] := ca[f + g, j, k, l] // S;
ca /: g_ * ca[f_, j_, k_, l_] := ca[g f, j, k, l] // S;
δaa[0, _, _, _, _] := 0;
δaa /: δaa[f_, i_, j_, k_, l_] + δaa[g_, i_, j_, k_, l_] :=
  δaa[f + g, i, j, k, l] // S;
δaa /: g_ * δaa[f_, i_, j_, k_, l_] := δaa[g f, i, j, k, l] // S;
```

```
β1[0] := 0;
β1 /: β1[f_] + β1[g_] := β1[f + g] // S;
β1 /: g_ * β1[f_] := β1[g f] // S;
a1[0, _, _] := 0;
a1 /: a1[f_, j_, k_] + a1[g_, j_, k_] := a1[f + g, j, k] // S;
a1 /: g_ * a1[f_, j_, k_] := a1[g f, j, k] // S;
c1[0, _] := 0;
c1 /: c1[f_, j_] + c1[g_, j_] := c1[f + g, j] // S;
c1 /: g_ * c1[f_, j_] := c1[g f, j] // S;
δa1[0, _, _] := 0;
δa1 /: δa1[f_, j_, k_] + δa1[g_, j_, k_] := δa1[f + g, j, k] // S;
δa1 /: g_ * δa1[f_, j_, k_] := δa1[g f, j, k] // S;
ca1[0, _, _, _] := 0;
ca1 /: ca1[f_, j_, k_, l_] + ca1[g_, j_, k_, l_] := ca1[f + g, j, k, l] // S;
ca1 /: g_ * ca1[f_, j_, k_, l_] := ca1[g f, j, k, l] // S;
δaa1[0, _, _, _, _] := 0;
δaa1 /: δaa1[f_, i_, j_, k_, l_] + δaa1[g_, i_, j_, k_, l_] :=
  δaa1[f + g, i, j, k, l] // S;
δaa1 /: g_ * δaa1[f_, i_, j_, k_, l_] := δaa1[g f, i, j, k, l] // S;
```

δaa relations

```
VS = False;
```

“First sort tails then sort heads”

Standard Swinging - sorts heads, if support is 4 strands:

```
S[δaa[f_, i_, j_, k_, l_]] /; DQ[i, j, k, l] ⋀ OQ[i, k] ⋀ ! OQ[j, l] := (
   If[VS, Print["Standard swinging on ", δaa[f, i, j, k, l]]];
   S[δaa[f, i, l, k, j] + ca[b_k f, l, i, j] -
     ca[b_i f, l, k, j] - ca[b_k f, j, i, l] + ca[b_i f, j, k, l]]
  );
```

Locality - sorts tails when supports are disjoint:

```
S[δaa[f_, i_, j_, k_, l_]] /; ({i, j} ⋂ {k, l} === {}) ⋀ ! OQ[i, k] := (
   If[VS, Print["Locality on ", δaa[f, i, j, k, l]]];
   δaa[f, k, l, i, j] // S
  );
```

Commute Heads - sorts tails when the heads are the same:

```
S[δaa[f_, i_, k_, j_, k_]] /; DQ[i, j, k] ⋀ ! OQ[i, j] := (
   If[VS, Print["Commute heads on ", δaa[f, i, k, j, k]]];
   S[δaa[f, j, k, i, k] + δa[-b_i f, j, k] + δa[b_j f, i, k]]
  );
```

Commute Head/Tail - sorts tails:

```
S[δaa[f_, i_, j_, k_, i_]] /; DQ[i, j, k] ⋀ ! OQ[i, k] := (
   If[VS, Print["Commute head/tail on ", δaa[f, i, j, k, i]]];
   S[
    δaa[f, k, i, i, j] + δaa[f, k, j, i, j] - δaa[f, i, j, k, j]
   ]
  );
```

Commute Head/Tail - sorts heads where heads & tails are both broken:

```
S[δaa[f_, k_, j_, j_, i_]] /; DQ[i, j, k] ⋀ OQ[i, j, k] := (
   If[VS, Print["Commute head/tail on ", δaa[f, k, j, j, i]]];
   S[
    δaa[f, j, i, k, j] + δaa[f, j, i, k, i] - δaa[f, k, i, j, i]
   ]
  );
```

2113 Swinging - sorts tails:

```
δaa[f, j, i, ii, k] // S
```

```
Locality on δaa[f, j, i, ii, k]

Standard swinging on δaa[f, ii, k, j, i]
```

$$\mathrm{ca}[-f\,b_{ii}, i, j, k] + \mathrm{ca}[f\,b_{ii}, k, j, i] + \mathrm{ca}[-f\,b_j, k, ii, i] + \mathrm{ca}[f\,b_j, i, ii, k] + \delta\mathrm{aa}[f, ii, i, j, k]$$

```
S[δaa[f_, j_, i_, i_, k_]] /; DQ[i, j, k] ⋀ OQ[i, j, k] := (
   If[VS, Print["2113 swinging on ", δaa[f, j, i, i, k]]];
   S[ca[-f b_i, i, j, k] + ca[f b_i, k, j, i] +
     c[-f b_j, k] ** aop[1, i] + ca[f b_j, i, i, k] + aop[f, i] ** δa[1, j, k]]
  );
```

```
δaa[f, j, i, i, k] // S
```

$$\mathrm{c}[f\,b_i\,b_j, k] + \mathrm{ca}[-f\,b_i, i, j, k] + \mathrm{ca}[f\,b_i, k, j, i] + \mathrm{ca}[-f\,b_j, k, i, i] + \mathrm{ca}[f\,b_j, i, i, k] + \delta\mathrm{a}[-f\,b_i, j, k] + \delta\mathrm{aa}[f, i, i, j, k]$$

3112 Swinging - sorts tails:

```
δaa[f, k, i, ii, j] // S
```

```
Locality on δaa[f, k, i, ii, j]

Standard swinging on δaa[f, ii, j, k, i]
```

$$\mathrm{ca}[-f\,b_{ii}, i, k, j] + \mathrm{ca}[f\,b_{ii}, j, k, i] + \mathrm{ca}[-f\,b_k, j, ii, i] + \mathrm{ca}[f\,b_k, i, ii, j] + \delta\mathrm{aa}[f, ii, i, k, j]$$

```
S[δaa[f_, k_, i_, i_, j_]] /; DQ[i, j, k] ⋀ OQ[i, j, k] := (
   If[VS, Print["3112 swinging on ", δaa[f, k, i, i, j]]];
   S[ca[-f b_i, i, k, j] + ca[f b_i, j, k, i] +
     c[-f b_k, j] ** aop[1, i] + ca[f b_k, i, i, j] + aop[1, i] ** δa[f, k, j]]
  );
```

```
δaa[f, k, i, i, j] // S
```

```
c[f b_i b_k, j] + ca[-f b_i, i, k, j] + ca[f b_i, j, k, i] +
 ca[-f b_k, j, i, i] + ca[f b_k, i, i, j] + δa[-f b_i, k, j] + δaa[f, i, i, k, j]
```

Tails Commute - sorts heads when the tails are the same:

```
S[δaa[f_, i_, j_, i_, l_]] /; DQ[i, j, l] ⋀ ! OQ[j, l] := (
   If[VS, Print["Tails commute on ", δaa[f, i, j, i, l]]];
   δaa[f, i, l, i, j] // S
  );
```

1321 Swinging - sorts heads:

```
δaa[f, i, k, j, ii] // S
```

```
Standard swinging on δaa[f, i, k, j, ii]
```

```
ca[-f b_i, ii, j, k] + ca[f b_i, k, j, ii] +
 ca[-f b_j, k, i, ii] + ca[f b_j, ii, i, k] + δaa[f, i, ii, j, k]
```

```
S[δaa[f_, i_, k_, j_, i_]] /; DQ[i, j, k] ⋀ OQ[i, j, k] := (
   If[VS, Print["1321 swinging on ", δaa[f, i, k, j, i]]];
   S[ca[-f b_i, i, j, k] + ca[f b_i, k, j, i] +
     ca[-f b_j, k, i, i] + ac[f b_j, i, k, i] + δaa[f, i, i, j, k]]
  );
```

1322 Swinging - sorts heads, but breaks tails:

```
δaa[f, i, k, j, jj] // S
```

```
Standard swinging on δaa[f, i, k, j, jj]
```

```
ca[-f b_i, jj, j, k] + ca[f b_i, k, j, jj] +
 ca[-f b_j, k, i, jj] + ca[f b_j, jj, i, k] + δaa[f, i, jj, j, k]
```

```
S[δaa[f_, i_, k_, j_, j_]] /; DQ[i, j, k] ⋀ OQ[i, j, k] := (
   If[VS, Print["1322 swinging on ", δaa[f, i, k, j, j]]];
   S[ac[-f b_i, j, k, j] + ca[f b_i, k, j, j] +
     ca[-f b_j, k, i, j] + ca[f b_j, j, i, k] + δaa[f, j, k, i, j]]
  );
```

1332 Swinging - sorts heads:

```
δaa[f, i, k, kk, j] // S
```

```
Standard swinging on δaa[f, i, k, kk, j]
```

```
ca[-f b_i, j, kk, k] + ca[f b_i, k, kk, j] +
 ca[-f b_kk, k, i, j] + ca[f b_kk, j, i, k] + δaa[f, i, j, kk, k]
```

```
S[δaa[f_, i_, k_, k_, j_]] /; DQ[i, j, k] ∧ OQ[i, j, k] := (
   If[VS, Print["1332 swinging on ", δaa[f, i, k, k, j]]];
   S[c[-f b_i, j] ** aop[1, k] + ca[f b_i, k, k, j] +
     ca[-f b_k, k, i, j] + ca[f b_k, j, i, k] + δa[f, i, j] ** aop[1, k]]
  );
```

```
δaa[f, i, k, k, j] // S
```

```
c[f b_i b_k, j] + ca[-f b_i, j, k, k] + ca[f b_i, k, k, j] +
 ca[-f b_k, k, i, j] + ca[f b_k, j, i, k] + δa[-f b_k, i, j] + δaa[f, i, j, k, k]
```

1231 Swinging - sorts heads:

```
δaa[f, i, j, k, ii] // S
```

```
Standard swinging on δaa[f, i, j, k, ii]
```

```
ca[-f b_i, ii, k, j] + ca[f b_i, j, k, ii] +
 ca[-f b_k, j, i, ii] + ca[f b_k, ii, i, j] + δaa[f, i, ii, k, j]
```

```
S[δaa[f_, i_, j_, k_, i_]] /; DQ[i, j, k] ∧ OQ[i, j, k] := (
   If[VS, Print["1231 swinging on ", δaa[f, i, j, k, i]]];
   S[ac[-f b_i, k, j, i] + ca[f b_i, j, k, i] +
     ca[-f b_k, j, i, i] + ac[f b_k, i, j, i] + δaa[f, i, i, k, j]]
  );
```

1211 sliding - sorts heads:

```
S[δaa[f_, i_, j_, i_, i_]] /; DQ[i, j] ∧ OQ[i, j] := (
   If[VS, Print["1211 sliding on ", δaa[f, i, j, i, i]]];
   S[δaa[f, i, i, i, j]]
  );
```

2111 sliding - sorts tails:

```
S[δaa[f_, j_, i_, i_, i_]] /; DQ[i, j] ∧ OQ[i, j] := (
   If[VS, Print["2111 sliding on ", δaa[f, j, i, i, i]]];
   S[δaa[f, i, i, j, i]]
  );
```

2212 sliding - sorts tails:

```
S[δaa[f_, j_, j_, i_, j_]] /; DQ[i, j] ∧ OQ[i, j] := (
   If[VS, Print["2212 sliding on ", δaa[f, j, j, i, j]]];
   S[δaa[f, i, j, j, j]]
  );
```

2221 sliding - sorts heads:

```
S[δaa[f_, j_, j_, j_, i_]] /; DQ[i, j] ⋀ OQ[i, j] := (
   If[VS, Print["2221 sliding on ", δaa[f, j, j, j, i]]];
   S[δaa[f, j, i, j, j]]
  );
```

2231 Swinging - sorts heads:

```
δaa[f, j, jj, k, i] // S
```

```
Standard swinging on δaa[f, j, jj, k, i]
```

```
ca[-f b_j, i, k, jj] + ca[f b_j, jj, k, i] +
 ca[-f b_k, jj, j, i] + ca[f b_k, i, j, jj] + δaa[f, j, i, k, jj]
```

```
S[δaa[f_, j_, j_, k_, i_]] /; DQ[i, j, k] ⋀ OQ[i, j, k] := (
   If[VS, Print["2231 swinging on ", δaa[f, j, j, k, i]]];
   S[ca[-f b_j, i, k, j] + ca[f b_j, j, k, i] +
     ac[-f b_k, j, i, j] + ca[f b_k, i, j, j] + δaa[f, j, i, k, j]]
  );
```

2331 Swinging - sorts heads:

```
δaa[f, j, k, kk, i] // S
```

```
ca[-f b_j, i, kk, k] + ca[f b_j, k, kk, i] +
 ca[-f b_kk, k, j, i] + ca[f b_kk, i, j, k] + δaa[f, j, i, kk, k]
```

```
S[δaa[f_, j_, k_, k_, i_]] /; DQ[i, j, k] ⋀ OQ[i, j, k] := (
   If[VS, Print["2331 swinging on ", δaa[f, j, k, k, i]]];
   S[c[-f b_j, i] ** aop[1, k] + ca[f b_j, k, k, i] +
     ca[-f b_k, k, j, i] + ca[f b_k, i, j, k] + δa[f, j, i] ** aop[1, k]]
  );
```

Backie jkkj Swinging - sorts tails or heads:

```
δaa[f, j, k, kk, jj] // S
```

```
ca[-f b_j, jj, kk, k] + ca[f b_j, k, kk, jj] +
 ca[-f b_kk, k, j, jj] + ca[f b_kk, jj, j, k] + δaa[f, j, jj, kk, k]
```

```
S[δaa[f_, j_, k_, k_, j_]] /; DQ[j, k] := (
   If[VS, Print["Backie swinging on ", δaa[f, j, k, k, j]]];
   S[c[-f b_j, j] ** aop[1, k] + ca[f b_j, k, k, j] +
     ca[-f b_k, k, j, j] + ac[f b_k, j, k, j] + δa[f, j, j] ** aop[1, k]]
  );
```

NonCommutativeMultiply

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[0, _] = 0; NonCommutativeMultiply[_, 0] = 0;
NonCommutativeMultiply[x_, x_] = 0;
NonCommutativeMultiply[x_Plus, y_] := NonCommutativeMultiply[#, y] & /@ x;
NonCommutativeMultiply[x_, y_Plus] := NonCommutativeMultiply[x, #] & /@ y;

β[f_] ** a[g_, j_, k_] := a[f g, j, k];
β[f_] ** c[g_, j_] := c[f g, j];
c[g_, j_] ** β[f_] := c[f g, j];
β[f_] ** δa[g_, j_, k_] := δa[f g, j, k];
β[f_] ** ca[g_, i_, j_, k_] := ca[f g, i, j, k];
ca[g_, i_, j_, k_] ** β[f_] := ca[f g, i, j, k];
β[f_] ** δaa[g_, i_, j_, k_, l_] := δaa[f g, i, j, k, l];
δa[g_, j_, k_] ** β[f_] := δa[f g, j, k];
δ ** a[f_, i_, j_] := δa[f, i, j];
c[f_, i_] ** a[g_, j_, k_] := ca[f g, i, j, k];
a[f_, i_, j_] ** δa[g_, k_, l_] := δaa[f g, i, j, k, l];
δa[f_, i_, j_] ** a[g_, k_, l_] := δaa[f g, i, j, k, l];

δ ** _c = 0;
δ ** _δa = 0;
δ ** _ca = 0;
δ ** _δaa = 0;
_c ** _c = 0;
_c ** _δa = _δa ** _c = 0;
_c ** _ca = _ca ** _c = 0;
_c ** _δaa = _δaa ** _c = 0;
_δa ** _δa = 0;
_δa ** _δaa = _δaa ** _δa = 0;
_δa ** _ca = _ca ** _δa = 0;

NonCommutativeMultiply::ndef =
  "NonCommuatativeMultiply is not defined on {`1`,`2`}."
NonCommutativeMultiply[x_, y_] :=
   (Message[NonCommutativeMultiply::ndef, x, y]; Undefined);
```

NonCommuatativeMultiply is not defined on {`1`,`2`}.

Bracket Generalities

```
B[0, _] = 0; B[_, 0] = 0;
B[x_, x_] = 0;
B[x_Plus, y_] := B[#, y] & /@ x;
B[x_, y_Plus] := B[x, #] & /@ y;
```

The γ shortcuts

```
γ[f_, j_, k_] := δa[f, j, k] - c[b_j f, k] // S;
γ[f_, j_, k_, l_] /; DQ[j, k, l] := ca[f, l, j, k] - ca[f, k, j, l] // S;
ac[f_, j_, k_, l_] := ca[f, l, j, k] + B[a[1, j, k], c[f, l]];
aop[f_, j_] := a[f, j, j] + β[-f b_j] + c[-f, j];
```

Fundamental Brackets

a-β, a-c, a-a, AS

```
B[a[j_, k_], β[g_]] := γ[∂_{b_j} g - ∂_{b_k} g, j, k];
B[β[g_], a[j_, k_]] := -B[a[j, k], β[g]];
B[a[j_, k_], a[l_, m_]] /; ({j, k} ⋂ {l, m} === {}) := 0;
B[a[j_, k_], a[j_, l_]] /; DQ[j, k, l] := γ[1, j, k, l] // S;
B[a[j_, k_], a[i_, k_]] /; DQ[i, j, k] := a[b_i, j, k] - a[b_j, i, k] // S;
B[a[j_, k_], a[k_, l_]] /; DQ[j, k, l] := a[b_j, k, l] - a[b_k, j, l] - γ[1, j, k, l] // S;
B[a[k_, l_], a[j_, k_]] /; DQ[j, k, l] := -B[a[j, k], a[k, l]];
(* backie *) B[a[j_, k_], a[k_, j_]] /; DQ[j, k] :=
  a[b_j, k, j] - a[b_k, j, k] + a[b_j, k, k] - a[b_k, j, j] + ca[1, k, k, j] -
   ca[1, j, j, k] + ca[1, k, j, j] - ca[1, j, k, k] + γ[1, j, k] - γ[1, k, j];
(* [tail, selfie] *) B[a[j_, k_], a[j_, j_]] /; DQ[j, k] := γ[1, j, k] // S;
B[a[j_, j_], a[j_, k_]] /; DQ[j, k] := -B[a[j, k], a[j, j]];
(* [head, selfie] *) B[a[j_, k_], a[k_, k_]] /; DQ[j, k] := γ[-1, j, k] // S;
B[a[k_, k_], a[j_, k_]] /; DQ[j, k] := -B[a[j, k], a[k, k]];
B[a[f_, j_, k_], c[g_, j_]] /; DQ[j, k] := γ[-f g, j, k];
B[a[f_, j_, k_], c[g_, k_]] /; DQ[j, k] := γ[f g, j, k];
B[a[f_, j_, k_], c[g_, l_]] /; ({j, k} ⋂ {l} === {}) := 0;
B[a[f_, j_, j_], c[g_, j_]] = 0;
B[c[g_, l_], a[f_, j_, k_]] := -B[a[f, j, k], c[g, l]];
```

Vanishing brackets

```
B[_β, _β | δ | _c | _δa | _ca | _δaa] = 0;
B[_β | δ | _c | _δa | _ca | _δaa, _β] = 0;
B[δ | _c | _δa | _ca | _δaa, δ | _c | _δa | _ca | _δaa] = 0;
```

Composite Brackets

```
B[a[f_, j_, k_], β[g_]] := β[f] ** B[a[j, k], β[g]];
B[β[g_], a[f_, j_, k_]] := -B[a[f, j, k], β[g]];
B[a[f_, j_, k_], a[l_, m_]] :=
  B[β[f], a[l, m]] ** a[1, j, k] + β[f] ** B[a[j, k], a[l, m]];
B[a[f_, j_, k_], a[g_, l_, m_]] :=
  B[a[f, j, k], β[g]] ** a[1, l, m] + β[g] ** B[a[f, j, k], a[l, m]];
B[a[f_, i_, j_], δa[g_, k_, l_]] := δ ** B[a[f, i, j], a[g, k, l]];
B[δa[f_, i_, j_], a[g_, k_, l_]] := δ ** B[a[f, i, j], a[g, k, l]];
B[a[f_, i_, j_], ca[g_, k_, l_, m_]] :=
  B[a[f, i, j], c[g, k]] ** a[1, l, m] + c[g, k] ** B[a[f, i, j], a[l, m]];
B[ca[g_, k_, l_, m_], a[f_, i_, j_]] := -B[a[f, i, j], ca[g, k, l, m]];
B[a[f_, i_, j_], δaa[g_, k_, l_, m_, n_]] :=
  B[a[f, i, j], δa[g, k, l]] ** a[1, m, n] + δa[g, k, l] ** B[a[f, i, j], a[m, n]];
B[δaa[g_, k_, l_, m_, n_], a[f_, i_, j_]] := -B[a[f, i, j], δaa[g, k, l, m, n]];

B::ndef = "B is not defined on {`1`,`2`}."
B[x_, y_] := (Message[B::ndef, x, y]; Undefined);

B is not defined on {`1`,`2`}.
```

The MM Presentation

U2MM

On the elements β, a, c, δa, ca, δaa.

```
β[f_] ** 𝟙 := β1[f];
a[f_, j_, k_] ** 𝟙 /; DQ[j, k] := a1[f, j, k];
a[f_, j_, j_] ** 𝟙 := a1[f, j, j] + δa1[-∂_{b_j} f, j, j] + c1[f + b_j ∂_{b_j} f, j];
c[f_, j_] ** 𝟙 := c1[f, j];
δa[f_, j_, k_] ** 𝟙 := δa1[f, j, k];
ca[f_, i_, j_, k_] ** 𝟙 /; DQ[i, j] := ca1[f, i, j, k];
ca[f_, j_, j_, k_] ** 𝟙 := ca1[f, j, j, k] + δa1[f, j, k] + c1[-b_j f, k];
δaa[f_, i_, j_, k_, l_] ** 𝟙 /; DQ[j, k] := δaa1[f, i, j, k, l];
δaa[f_, i_, j_, j_, k_] ** 𝟙 :=
  δaa1[f, i, j, j, k] + δaa1[f, j, k, i, k] - δaa1[f, i, k, j, k];
```

MM2U

```
MM2U::ndef = "MM2U is not defined on `1`.";
MM2U[expr_] := expr /. {
   β1[f_] :→ β[f],
   a1[f_, j_, k_] /; DQ[j, k] :→ a[f, j, k],
   a1[f_, j_, j_] :→ a[f, j, j] + δa[∂_{b_j} f, j, j] + c[-f - b_j ∂_{b_j} f, j],
   c1[f_, j_] :→ c[f, j],
   δa1[f_, j_, k_] :→ δa[f, j, k],
   ca1[f_, i_, j_, k_] /; DQ[i, j] :→ ca[f, i, j, k],
   ca1[f_, j_, j_, k_] :→ ca[f, j, j, k] + δa[-f, j, k] + c[b_j f, k],
   δaa1[f_, i_, j_, k_, l_] /; DQ[j, k] :→ δaa[f, i, j, k, l],
   δaa1[f_, i_, i_, i_, j_] :→ δaa[f, i, j, i, i],
   δaa1[f_, i_, j_, j_, k_] /; DQ[i, j, k] :→
    δaa[f, i, j, j, k] - δaa[f, j, k, i, k] + δaa[f, i, k, j, k],
   δaa1[f_, i_, j_, j_, j_] /; DQ[i, j] :→
    δaa[f, j, j, i, j] + δa[b_j f, i, j] - δa[b_i f, j, j],
   δaa1[f_, i_, j_, j_, i_] /; DQ[i, j] :→ δaa[f, i, j, j, i] +
     δaa[f, i, i, j, i] - δaa[f, i, i, j, i] - δa[b_i f, j, i] + δa[b_j f, i, i],
   δaa1[other___] :→ (Message[MM2U::ndef, δaa1[other]]; Undefined)
  }
```

S for MM

```
VS1 = True;
S[β1[f_]] := β1[Simp[f]];
S[a1[i_, j_]] := a1[i, j];
S[a1[f_, i_, j_]] := a1[Simp[f], i, j];
S[c1[f_, k_]] := c1[Simp[f], k];
S[δa1[f_, i_, j_]] := δa1[Simp[f], i, j];
S[ca1[f_, j_, k_, l_]] := ca1[Simp[f], j, k, l];
S[δaa1[f_, i_, j_, k_, l_]] := δaa1[Simp[f], i, j, k, l];
(* Bad heads - interchange arrows no matter what: *)
S[δaa1[f_, i_, j_, k_, l_]] /; !OQ[j, l] := (
   If[VS1, Print["Interchange arrows on ", δaa1[f, i, j, k, l]]];
   S[δaa1[f, k, l, i, j]]
  );
(* Bad tails and good heads - swing: *)
S[δaa1[f_, i_, j_, k_, l_]] /; !OQ[i, k] ∧ DQ[j, l] ∧ OQ[j, l] := (
   If[VS1, Print["Swinging on ", δaa1[f, i, j, k, l]]];
   S[δaa1[f, i, l, k, j] + ca1[b_k f, l, i, j] -
     ca1[b_i f, l, k, j] - ca1[b_k f, j, i, l] + ca1[b_i f, j, k, l]]
  );
(* Bad tails and equal heads - commute heads: *)
S[δaa1[f_, i_, k_, j_, k_]] /; !OQ[i, j] := (
   If[VS1, Print["Commute heads on ", δaa1[f, i, k, j, k]]];
   S[δaa1[f, j, k, i, k] + δa1[-b_i f, j, k] + δa1[b_j f, i, k]]
  );
```

Bases

```
FormalBasis[S_List, f_] := Module[{ff, n = Length@S, i, j, k, l},
   ff = f @@ Table[b_S[[i]], {i, n}];
   Flatten@{
      β[ff],
      Table[a[ff, S[[i]], S[[j]]], {i, n}, {j, n}],
      Table[c[ff, S[[i]]], {i, n}],
      Table[δa[ff, S[[i]], S[[j]]], {i, n}, {j, n}],
      Table[ca[ff, S[[i]], S[[j]], S[[k]]], {i, n}, {j, n}, {k, n}],
      Table[δaa[ff, S[[i]], S[[j]], S[[k]], S[[l]]], {i, n}, {j, n}, {k, i, n}, {l, j, n}]
     } /. 1[___] → 1
  ];
FormalBasis[n_Integer, f_] := FormalBasis[Range[n], f];
FormalPlusBasis[n_, f_] := Module[{ff},
   ff = f @@ Table[b_i, {i, n}];
   Flatten@{
      β[ff],
      Table[a[ff, i, j], {i, n - 1}, {j, i + 1, n}],
      Table[c[ff, i], {i, n}],
      Table[δa[ff, i, j], {i, n - 1}, {j, i + 1, n}],
      Table[ca[ff, i, j, k], {i, n}, {j, n - 1}, {k, j + 1, n}],
      Table[δaa[ff, i, j, k, l], {i, n - 1}, {j, i + 1, n}, {k, n - 1}, {l, k + 1, n}]
     } /. 1[___] → 1
  ];

Jacobi[x1_, x2_, x3_] := Module[{Jac},
   Jac = S[B[x1, B[x2, x3]] + B[x2, B[x3, x1]] + B[x3, B[x1, x2]]];
   If[Jac === 0, Jac, {x1, x2, x3} → Jac]
  ];
```

The Adjoint action

AutoAd

```
AutoAd[x_][y_] :=
  Module[{pows, states, i, s, seq, sh = 5, dseq, sf1, sf2, sf, t1, n},
   pows = NestList[B[x, #] &, y, 20];
   Print["pows computed for ", {x, y}, "..."];
   states = Union[
     Cases[pows, s_β | s_a | s_c | s_δa | s_ca | s_δaa ⧴ ReplacePart[s, 1 → _], ∞]];
   Sum[
    seq = Cases[{#}, states〚i〛, ∞] & /@ pows;
    seq = Replace[seq, {{_[f_, ___]} ⧴ f, {} → 0}, {1}];
    Print["seq computed... ", states〚i〛, " is ", i, "/", Length@states];
    dseq = Drop[seq, sh];
    If[Union[Length[MonomialList[#]] & /@ dseq] === {1} ⋀
      Union[Length[FactorTermsList[#]] & /@ dseq] === {2},
     sf1 = FindSequenceFunction[FactorTermsList[#]〚1〛 & /@ dseq];
     sf2 = FindSequenceFunction[FactorTermsList[#]〚2〛 & /@ dseq];
     Print["sf1: ", sf1, " sf2: ", sf2];
     sf = (sf1[#] sf2[#] &),
     (*Else*) sf = FindSequenceFunction[dseq,
       FunctionSpace → {"ConstantRecursive", "HolonomicSequence",
         "Polynomial", "RationalFunction", "HypergeometricTerm"}];
     Print["sf: ", sf];
    ];
    ReplacePart[states〚i〛, 1 → Simplify[Sum[seq〚n + 1〛/n!, {n, 0, sh-1}] + Sum[sf[n + 1 - sh]/n!, {n, sh, ∞}]]],
    {i, Length@states}]
  ];
(* Hint: Perhaps improve using Variables, CoefficientList, FromCoefficientList *)
```

Ad

```
Ad[a[t_, j_, k_]][β[f_]] /; FreeQ[t, b_] :=
  β[f] + c[(1 - ⅇ^(-t b_j)) (∂_{b_k} f - ∂_{b_j} f), k] + δa[((ⅇ^(-t b_j) - 1) (∂_{b_k} f - ∂_{b_j} f))/b_j, j, k];
Ad[a[t_, j_, k_]][a[1, j_, k_]] /; FreeQ[t, b_] := a[1, j, k];
Ad[a[t_, j_, k_]][a[1, n_, i_]] /;
   FreeQ[t, b_] ⋀ ({j, k} ⋂ {n, i} === {}) := a[1, n, i];
Ad[a[t_, j_, k_]][a[1, j_, j_]] /; DQ[j, k] ⋀ FreeQ[t, b_] :=
```

$$a[1, j, j] + c\left[-1 + e^{-t b_j}, k\right] + \delta a\left[\frac{1 - e^{-t b_j}}{b_j}, j, k\right];$$

Ad[a[t_, j_, k_]][a[1, k_, k_]] /; DQ[j, k] ⋀ FreeQ[t, b_] :=

$$a[1, k, k] + c\left[1 - e^{-t b_j}, k\right] + \delta a\left[\frac{-1 + e^{-t b_j}}{b_j}, j, k\right];$$

Ad[a[t_, j_, k_]][a[1, i_, j_]] /; DQ[i, j, k] ⋀ FreeQ[t, b_] :=

$$a[1, i, j] + a\left[1 - e^{-t b_j}, i, k\right] + a\left[\frac{\left(e^{-t b_j} - 1\right) b_i}{b_j}, j, k\right] + ca\left[\frac{1 - e^{-t b_j}}{b_j}, k, i, j\right] + ca\left[\frac{e^{-t b_j} - 1}{b_j}, j, i, k\right] + ca\left[\frac{b_i \left(1 - e^{-t b_j} - t b_j\right)}{b_j^2}, j, j, k\right] + ca\left[\frac{e^{-2 t b_j} b_i \left(1 - e^{t b_j} - e^{t b_j} \left(e^{t b_j} - 2\right) t b_j\right)}{b_j^2}, k, j, k\right] + ca\left[\frac{e^{-2 t b_j} \left(e^{t b_j} \left(1 - t b_j\right) - 1\right)}{b_j}, k, i, k\right] + \delta a\left[\frac{b_i \left(1 - e^{-t b_j} - t b_j\right)}{b_j^2} + \frac{-b_i \left(1 - e^{-2 t b_j} + \left(-1 - e^{-t b_j}\right) t b_j\right)}{b_j^2}, j, k\right] + \delta a\left[\frac{\left(-1 + e^{-t b_j} + t b_j\right)}{b_j} + \frac{\left(1 - e^{-2 t b_j} + \left(-1 - e^{-t b_j}\right) t b_j\right)}{b_j}, i, k\right] + \delta aa\left[\frac{2 e^{-t b_j} b_i (\mathrm{Sinh}[t b_j] - t b_j)}{b_j^3}, j, k, j, k\right] + \delta aa\left[\frac{-1 + e^{-t b_j} + t b_j}{b_j^2}, i, j, j, k\right] + \delta aa\left[-\frac{1 - e^{-2 t b_j} + \left(-1 - e^{-t b_j}\right) t b_j}{b_j^2}, i, k, j, k\right];$$

Ad[a[t_, j_, k_]][a[1, i_, k_]] /; DQ[i, j, k] ⋀ FreeQ[t, b_] :=

$$a\left[e^{-t b_j}, i, k\right] + a\left[\frac{\left(1 - e^{-t b_j}\right) b_i}{b_j}, j, k\right] + ca\left[\frac{2 e^{-t b_j} b_i (\mathrm{Sinh}[t b_j] - t b_j)}{b_j^2}, k, j, k\right] + ca\left[\frac{e^{-2 t b_j} \left(1 + e^{t b_j} \left(-1 + t b_j\right)\right)}{b_j}, k, i, k\right] + \delta a\left[\frac{e^{-2 t b_j} b_i \left(-1 + e^{t b_j} \left(1 - t b_j\right)\right)}{b_j^2}, j, k\right] + \delta a\left[\frac{e^{-2 t b_j} \left(1 - e^{t b_j} \left(1 - t b_j\right)\right)}{b_j}, i, k\right] + \delta aa\left[\frac{2 e^{-t b_j} b_i (-\mathrm{Sinh}[t b_j] + t b_j)}{b_j^3}, j, k, j, k\right] + \delta aa\left[\frac{e^{-2 t b_j} \left(-1 + e^{t b_j} \left(1 - t b_j\right)\right)}{b_j^2}, i, k, j, k\right];$$

Ad[a[t_, j_, k_]][a[1, j_, l_]] /; DQ[j, k, l] ⋀ FreeQ[t, b_] :=

$$a[1, j, l] + ca[t, l, j, k] + ca\left[\frac{e^{-t b_j} - 1}{b_j}, k, j, l\right] + \delta aa\left[\frac{1 - e^{-t b_j} - t b_j}{b_j^2}, j, k, j, l\right];$$

Ad[a[t_, j_, k_]][a[1, k_, l_]] /; DQ[j, k, l] ⋀ FreeQ[t, b_] :=

$$a\left[e^{t b_j}, k, l\right] + a\left[-\frac{\left(-1 + e^{t b_j}\right) b_k}{b_j}, j, l\right] + ca\left[\frac{-1 + e^{t b_j} \left(1 - t b_j\right)}{b_j}, k, k, l\right] + ca\left[\frac{b_j - e^{-t b_j} b_j + b_k + e^{t b_j} \left(-1 + t b_j\right) b_k}{b_j^2}, k, j, l\right] + ca\left[\frac{b_j + b_k + t b_j b_k - e^{t b_j} (b_j + b_k)}{b_j^2}, l, j, k\right] + \delta aa\left[\frac{1 + e^{t b_j} \left(-1 + t b_j\right)}{b_j^2}, j, k, k, l\right] +$$

```
δaa[(1/b_j^3) e^(-t b_j) (b_j + e^(2 t b_j) (b_j + (2 - t b_j) b_k) - e^(t b_j) (2 b_k + b_j (2 + t b_k))), j, k, j, l];
Ad[a[t_, j_, k_]][a[1, k_, j_]] /; DQ[j, k] ∧ FreeQ[t, b_] :=
  a[e^(t b_j), k, j] + a[-1 + e^(t b_j), k, k] + a[-((-1 + e^(t b_j)) b_k)/b_j, j, j] + a[-((-1 + e^(t b_j)) b_k)/b_j, j, k] +
   c[((1 - e^(t b_j) + 2 e^(t b_j) t b_j) b_k)/b_j, j] + c[-1 + e^(-t b_j) + (e^(-t b_j) (1 - e^(2 t b_j) + 2 e^(2 t b_j) t b_j) b_k)/b_j, k] +
   ca[e^(t b_j) t, k, k, j] + ca[(-1 + e^(t b_j) - 2 e^(t b_j) t b_j)/b_j, j, k, k] + ca[(-1 + e^(t b_j) - e^(t b_j) t b_j)/b_j, k, k, k] +
   ca[(-2 (-1 + e^(t b_j)) b_k + b_j (1 - e^(t b_j) + 2 e^(t b_j) t b_k))/b_j^2, j, j, k] +
   ca[(e^(-t b_j) ((-1 + e^(t b_j)) b_k + b_j (-1 + e^(t b_j) - e^(2 t b_j) t b_k)))/b_j^2, k, j, j] +
   ca[(1/b_j^2) e^(-t b_j) (-(1 - 3 e^(t b_j) + 2 e^(2 t b_j)) b_k + b_j (-(-1 + e^(t b_j))^2 + e^(2 t b_j) t b_k)), k, j, k] +
   δa[-e^(t b_j) t, k, j] + δa[(-1 + e^(t b_j) - e^(t b_j) t b_j)/b_j, k, k] + δa[-((1 - e^(t b_j) + e^(t b_j) t b_j) b_k)/b_j^2, j, j] +
   δa[(e^(-t b_j) ((-1 + e^(t b_j)) b_k + b_j (-1 + e^(t b_j) - e^(2 t b_j) t b_k)))/b_j^2, j, k] +
   δaa[(2 - 2 e^(t b_j) + 2 e^(t b_j) t b_j)/b_j^2, j, j, k, k] + δaa[(2 - 2 e^(t b_j) + 2 e^(t b_j) t b_j)/b_j^2, j, k, k, k] +
   δaa[(1/b_j^3) e^(-t b_j) ((1 - 4 e^(t b_j) + 3 e^(2 t b_j)) b_k + b_j ((-1 + e^(t b_j))^2 - 2 e^(2 t b_j) t b_k)), j, j, j, k] +
   δaa[(1/b_j^3) e^(-t b_j) ((1 - 4 e^(t b_j) + 3 e^(2 t b_j)) b_k + b_j ((-1 + e^(t b_j))^2 - 2 e^(2 t b_j) t b_k)), j, k, j, k];
Ad[a[t_, j_, k_]][c[1, i_]] /; FreeQ[t, b_] ∧ ({j, k} ⋂ {i} === {}) := c[1, i];
Ad[a[t_, j_, k_]][c[1, j_]] /; DQ[j, k] ∧ FreeQ[t, b_] :=
  c[1, j] + c[1 - e^(-t b_j), k] + δa[(e^(-t b_j) - 1)/b_j, j, k];
Ad[a[t_, j_, k_]][c[1, k_]] /; DQ[j, k] ∧ FreeQ[t, b_] :=
  c[e^(-t b_j), k] + δa[(1 - e^(-t b_j))/b_j, j, k];

Ad[x_β | x_c | x_δa | x_ca | x_δaa][y_] := y + B[x, y];

Ad[x_][a[f_, i_, j_]] /; f =!= 1 := Ad[x][β[f]] ** Ad[x][a[1, i, j]];
Ad[x_][c[f_, i_]] /; f =!= 1 := Ad[x][β[f]] ** Ad[x][c[1, i]];
Ad[x_][δa[f_, j_, k_]] := δ ** (β[f] ** Ad[x][a[1, j, k]]);
Ad[x_][ca[f_, i_, j_, k_]] := Ad[x][c[f, i]] ** Ad[x][a[1, j, k]];
Ad[x_][δaa[f_, i_, j_, k_, l_]] := Ad[x][δa[f, i, j]] ** Ad[x][a[1, k, l]];
Ad[x_][y_Plus] := Ad[x] /@ y;
```

```
Ad::ndef = "Ad[`1` is not defined on `2`.";
Ad[x_][y_] := (Message[Ad::ndef, x, y]; Undefined);
```

Verifying R3

```
VerifyR3[t_, expr_] := Module[{lhs, rhs},
   lhs = expr // R[t, 1, 2] // R[t, 1, 3] // R[t, 2, 3] // S;
   rhs = expr // R[t, 2, 3] // R[t, 1, 3] // R[t, 1, 2] // S;
   expr → S[lhs - rhs] == 0
  ];
VerifyR3[expr_] := VerifyR3[1, expr];
```

```
Total[MapIndexed[(#1 /. f → f#2[[1]]) &, DeleteCases[FormalBasis[{j, k}, f], _β | _a]]]
```

$c[f_1[b_j, b_k], j] + c[f_2[b_j, b_k], k] + ca[f_7[b_j, b_k], j, j, j] + ca[f_8[b_j, b_k], j, j, k] + ca[f_9[b_j, b_k], j, k, j] + ca[f_{10}[b_j, b_k], j, k, k] + ca[f_{11}[b_j, b_k], k, j, j] + ca[f_{12}[b_j, b_k], k, j, k] + ca[f_{13}[b_j, b_k], k, k, j] + ca[f_{14}[b_j, b_k], k, k, k] + \delta a[f_3[b_j, b_k], j, j] + \delta a[f_4[b_j, b_k], j, k] + \delta a[f_5[b_j, b_k], k, j] + \delta a[f_6[b_j, b_k], k, k] + \delta aa[f_{15}[b_j, b_k], j, j, j, j] + \delta aa[f_{16}[b_j, b_k], j, j, j, k] + \delta aa[f_{17}[b_j, b_k], j, j, k, j] + \delta aa[f_{18}[b_j, b_k], j, j, k, k] + \delta aa[f_{19}[b_j, b_k], j, k, j, k] + \delta aa[f_{20}[b_j, b_k], j, k, k, k] + \delta aa[f_{21}[b_j, b_k], k, j, k, j] + \delta aa[f_{22}[b_j, b_k], k, j, k, k] + \delta aa[f_{23}[b_j, b_k], k, k, k, k]$

```
f22[__] = 0; f21[__] = 0; f9[__] = 0; f5[__] = 0; f13[__] = 0; f17[__] = 0; f7[__] = 0;
f8[__] = 0;
f10[bj_, bk_] := g2[bk];
f1[bj_, bk_] := g3[bk];
f12[bj_, bk_] := -bj f19[bj, bk];
f19[bj_, bk_] := -(E^(-bj) (2 - 2 E^bj + bj + E^bj bj))/(2 bj^3);
f16[bj_, bk_] := g1[bj];
f11[bj_, bk_] := g4[bj];
f14[bj_, bk_] := -bj f20[bj, bk];
f20[bj_, bk_] := (-1 + E^bj)/(E^bj bj) (-2 + 2 E^bk - bk - E^bk bk + 2 bk^2 g2[bk] + 2 E^bk bk^2 g4[bk])/(2 (-1 + E^bk) bk^2);
f2[bj_, bk_] := g5[bj] - bj f4[bj, bk];
(* Non-forced choice: *) f4[__] = 0;
g5[bj_] := -(E^(-bj) (2 - 2 E^bj + bj + E^bj bj + 2 bj g3[bj]))/(2 bj);
(* Non-forced choices: *) g1[_] = 0; g2[_] = 0; g3[_] = 0;
g4[_] = 0; f3[__] = 0; f6[__] = 0; f15[__] = 0; f18[__] = 0; f23[__] = 0;
Total[
  MapIndexed[(#1 /. f -> f#2[[1]]) &, DeleteCases[FormalBasis[{j, k}, f], _β | _a]]] // S
```

$$c\left[-\frac{e^{-b_j}\left(2-2e^{b_j}+\left(1+e^{b_j}\right)b_j\right)}{2b_j},k\right]+ca\left[\frac{e^{-b_j}\left(2-2e^{b_j}+\left(1+e^{b_j}\right)b_j\right)}{2b_j^2},k,j,k\right]+$$
$$ca\left[\frac{e^{-b_j}\left(-1+e^{b_j}\right)\left(2-2e^{b_k}+\left(1+e^{b_k}\right)b_k\right)}{2\left(-1+e^{b_k}\right)b_k^2},k,k,k\right]+$$
$$\delta aa\left[-\frac{e^{-b_j}\left(2-2e^{b_j}+\left(1+e^{b_j}\right)b_j\right)}{2b_j^3},j,k,j,k\right]+$$
$$\delta aa\left[-\frac{e^{-b_j}\left(-1+e^{b_j}\right)\left(2-2e^{b_k}+\left(1+e^{b_k}\right)b_k\right)}{2\left(-1+e^{b_k}\right)b_j b_k^2},j,k,k,k\right]$$

```
R[t_, j_, k_][x_] := Expand[
   x // Ad[a[t, j, k]] //
    (# + t B[Total[MapIndexed[(#1 /. f -> f#2[[1]]) &, DeleteCases[
            FormalBasis[{j, k}, f], _β | _a]]], #]) &
  ];
R[j_, k_][x_] := R[1, j, k][x];
```

```
Print[VerifyR3[#]] & /@
  {a[f[b1, b2, b3, b4], 1, 4], a[f[b1, b2, b3, b4], 2, 4], a[f[b1, b2, b3, b4], 3, 4],
   a[f[b1, b2, b3, b4], 4, 1], a[f[b1, b2, b3, b4], 4, 2], a[f[b1, b2, b3, b4], 4, 3]};
```

```
a[f[b1, b2, b3, b4], 1, 4] → True

a[f[b1, b2, b3, b4], 2, 4] → True

a[f[b1, b2, b3, b4], 3, 4] → True

a[f[b1, b2, b3, b4], 4, 1] → True

a[f[b1, b2, b3, b4], 4, 2] → True

a[f[b1, b2, b3, b4], 4, 3] → True
```

```
Print[VerifyR3[#]] & /@
  {a[f[b1, b2, b3, b4], 1, 2], a[f[b1, b2, b3, b4], 1, 3], a[f[b1, b2, b3, b4], 2, 3],
   a[f[b1, b2, b3, b4], 2, 1], a[f[b1, b2, b3, b4], 3, 1], a[f[b1, b2, b3, b4], 3, 2]};
```

```
a[f[b1, b2, b3, b4], 1, 2] → True

a[f[b1, b2, b3, b4], 1, 3] → True

a[f[b1, b2, b3, b4], 2, 3] → True

a[f[b1, b2, b3, b4], 2, 1] → True

a[f[b1, b2, b3, b4], 3, 1] → True

a[f[b1, b2, b3, b4], 3, 2] → True
```

```
Print[VerifyR3[t, #]] & /@
  {a[f[b1, b2, b3, b4], 1, 4], a[f[b1, b2, b3, b4], 2, 4], a[f[b1, b2, b3, b4], 3, 4],
   a[f[b1, b2, b3, b4], 4, 1], a[f[b1, b2, b3, b4], 4, 2], a[f[b1, b2, b3, b4], 4, 3]};
```

```
a[f[b1, b2, b3, b4], 1, 4] → True

a[f[b1, b2, b3, b4], 2, 4] →
```

$$\mathrm{ca}\left[-\frac{1}{b_1}e^{-b_1-t\,b_2}\left(-1+e^{t\,b_2}\right)f[b_1,b_2,b_3,b_4]\left(2\left(e^{b_1}+e^{(1+t)\,b_1}(-1+t)-e^{t\,b_1}t\right)+\left(e^{b_1}-e^{t\,b_1}\right)t\,b_1\right),\ 3,2,4\right]+\mathrm{ca}\left[\frac{1}{b_1^2}e^{-b_1-t\,b_2}\left(-1+e^{t\,b_2}\right)f[b_1,b_2,b_3,b_4]\left(2\left(e^{b_1}+e^{(1+t)\,b_1}(-1+t)-e^{t\,b_1}t\right)+\left(e^{b_1}-e^{t\,b_1}\right)t\,b_1\right)b_2,\ 3,1,4\right]+$$
$$\delta\mathrm{aa}\left[\frac{1}{b_1^2}e^{-b_1-t\,b_2}\left(-1+e^{t\,b_2}\right)f[b_1,b_2,b_3,b_4]\left(2\left(e^{b_1}+e^{(1+t)\,b_1}(-1+t)-e^{t\,b_1}t\right)+\left(e^{b_1}-e^{t\,b_1}\right)t\,b_1\right),\ 1,3,2,4\right]+\delta\mathrm{aa}\left[-\frac{1}{b_1^3}e^{-b_1-t\,b_2}\left(-1+e^{t\,b_2}\right)f[b_1,b_2,b_3,b_4]\left(2\left(e^{b_1}+e^{(1+t)\,b_1}(-1+t)-e^{t\,b_1}t\right)+\left(e^{b_1}-e^{t\,b_1}\right)t\,b_1\right)b_2,\ 1,3,1,4\right]=0$$

```
$Aborted
```

```
ρ0 =
 Total[MapIndexed[(#1 /. f → f_{#2[[1]]}) &, DeleteCases[FormalBasis[{j, k}, f], _β | _a]]]
```

$$c\left[-\frac{e^{-b_j}\left(2-2e^{b_j}+b_j+e^{b_j}b_j\right)}{2b_j},k\right]+ca\left[\frac{e^{-b_j}\left(2-2e^{b_j}+b_j+e^{b_j}b_j\right)}{2b_j^2},k,j,k\right]+$$
$$ca\left[-\frac{e^{-b_j}\left(-1+e^{b_j}\right)\left(-2+2e^{b_k}-b_k-e^{b_k}b_k\right)}{2\left(-1+e^{b_k}\right)b_k^2},k,k,k\right]+$$
$$\delta aa\left[-\frac{e^{-b_j}\left(2-2e^{b_j}+b_j+e^{b_j}b_j\right)}{2b_j^3},j,k,j,k\right]+$$
$$\delta aa\left[\frac{e^{-b_j}\left(-1+e^{b_j}\right)\left(-2+2e^{b_k}-b_k-e^{b_k}b_k\right)}{2\left(-1+e^{b_k}\right)b_j b_k^2},j,k,k,k\right]$$

$$\phi_1[x\_] := e^{-x}-1;\ \phi_2[x\_] := \frac{(x+2)e^{-x}-2+x}{2x};$$

$$\rho = c[-\phi_2[b_j],k]+ca\left[\frac{\phi_2[b_j]}{b_j},k,j,k\right]+ca\left[\frac{\phi_1[b_j]}{b_k\,\phi_1[b_k]}\phi_2[b_k],k,k,k\right]+$$
$$\delta aa\left[\frac{-\phi_2[b_j]}{b_j^2},j,k,j,k\right]+\delta aa\left[\frac{-\phi_1[b_j]}{b_j\,b_k\,\phi_1[b_k]}\phi_2[b_k],j,k,k,k\right];$$

```
ρ0 - ρ
```

0